

Hierarchical Testbench Configuration Using `uvm_config_db`

June 2014

Authors

Hannes Nurminen
Professional
Services, Synopsys

Satya Durga Ravi
Professional
Services, Synopsys

Abstract

SoC designs have become extremely complex as more and more IP blocks are integrated into them. This increases the verification challenge manifold in terms of configuration and data handling, as well as architecting and maintaining a large verification environment. Hence it has become very important to create a robust and reusable testbench using a proven methodology that does not just facilitate but also improves the efficiency in verifying different configurations of the device under test (DUT).

Accellera Systems Initiative's Universal Verification Methodology (UVM), a stable and widely used methodology for architecting testbenches for verification of complex designs helps mitigate these verification challenges to create a scalable, robust and reusable test environment. UVM provides a vast set of macro, policy and base classes that help facilitate the creation of these testbenches, including an easy way to pass objects and variables across testbench hierarchy.

For engineers who are new to verification methodologies or are in the process of adopting UVM, this paper focuses on the UVM configuration mechanism "`uvm_config_db`", which helps in passing different class properties across hierarchy testbench components. Through the use of examples, the usage, techniques, and limitations of `uvm_config_db` are explained.

Introduction

To address the needs of today's verification architecture, a hierarchical setup of components is necessary to easily move or share configurations and other parameters across different testbench components. To enable this, UVM provides the infrastructure to maintain a database of objects and variables that can be populated and accessed using strings. This is achieved using the UVM syntax `uvm_config_db`.

Using `uvm_config_db`, objects can share the handle to their data members with other objects. Other testbench components can get access to the object without knowing where it exists in the hierarchy. It's almost like making some class variables global or public. Any testbench component can place handles and get handles to objects. In the database, handles are identified by assigned 'type' and 'name'.

Primarily there are two `uvm_config_db` functions, `set()` and `get()`. Any verification component using `set()` gives others access to the object it has created and controls which components have visibility to the object it has shared. The object can be shared globally or made available to one or more specific testbench components. Verification components using `get()` check if there is a shared handle matching the used parameters. The `get()` function defines the object type, the name and hierarchical path to the object searched for.

How to Use It – Different Syntax and Operation

Explicit `set()` and `get()` call functions are how you interact with the `uvm_config_db`. The `uvm_config_db` class functions are static, so they must be called using the "::" operator.

```

2 uvm_config_db#(<type>)::set(uvm_component cntxt,
3                               string      inst_name,
4                               string      field_name,
5                               <type>     value)
6
7 uvm_config_db#(<type>)::get(uvm_component cntxt,
8                               string      inst_name,
9                               string      field_name,
10                              ref         value)

```

Figure 1: set() and get() function syntax

“cntxt” and “inst_name” are used to specify the storage location or address of the object handle. When used properly these parameters define the hierarchical path to the object data.

“field_name” is the name for the object. It does not have to match the object’s actual name in the source code. Objects using **set()** and **get()** must use exactly the same name, otherwise the receiving party (**get()**) will fail to find the object from **uvm_config_db**.

“value” is the actual object handle shared through the **uvm_config_db**. Multiple recipients accessing an object via **get()**, will access the same object.

“<type>” is used as a parameter for the **uvm_config_db** class to identify the object from the **uvm_config_db**. “<type>” which may be either an integral or string, is the class name of the “value”. The exception is with enumerated type variables which must use **int** otherwise the **set()** won’t work as expected.

```

13 typedef enum {single,incr,wrap4,incr4,wrap8,incr8,wrap16,incr16}hburst_t
14 hburst_t      hburst;
15 uvm_config_db#(int)::set(this,"a","hburst",incr);

```

Figure 2: config_db for enum type

The **set()** specifies the “address” (cntxt & inst_name) where the object handle is stored to control the recipient(s) of the object. The **get()** has the same flexibility, and can freely select from where the information is to be fetched. In practice **get()** can be used to fetch an object destined to any component in the hierarchy. Typically for **set()** and **get()**, **this** is used in the “cntxt” field to specify the current instance/scope. **set()** uses “inst_name” to address the object to the appropriate sub-block in the hierarchy. **get()** often uses empty (“”) inst_name, since it typically is getting the objects destined for itself.

```

5 uvm_config_db#(int)::set(this,"my_subblock_a","max_cycles",max_cycles)
6 uvm_config_db#(int)::get(this,"", "max_cycles",max_cycles)

```

Figure 3: set() and get() typical use

uvm_config_db has two additional functions **exists()** and **wait_modified()**. **exists()** verifies that the defined variable is found in the **uvm_config_db**. The **wait_modified()** function blocks execution until the defined variable is accessed with the **set()** call.

```

2 uvm_config_db#(int)::exists(this,"my_subblock_a","max_cycles")
3 uvm_config_db#(int)::wait_modified(this,"my_subblock_a","max_cycles")

```

Figure 4: exists() and wait_modified() typical use

Automatic Configuration

UVM also offers build-time configuration of `uvm_component` (and extended) classes utilizing `uvm_config_db`. In automatic configuration, it is sufficient to call `set()` from an upper layer in the hierarchy and the `get()` will automatically execute at build time without requiring an explicit call. Automatic configuration utilizes the `uvm_config_db` feature “under the hood” to pass the configuration values from higher level testbench components in the hierarchy to its lower level components.

For automatic configuration to work there are two important requirements:

- ▶ The variable or object must have the appropriate FLAG in `uvm_field_*` macros
- ▶ `super()` must be called in `build_phase()`

```
3 class agents extends uvm_agent;
4   int i4;
5   `uvm_component_utils_begin (agent)
6     `uvm_field_int (i4, UVM_ALL_ON)
7   `uvm_component_utils_end
8
9   virtual function void build_phase(uvm_phase phase);
10    super.build_phase(phase);
11    ...
12  endfunction
13  ...
14 endclass
```

Once the component properties have the `uvm_field_*` declaration(s) in place with the appropriate FLAG(s), the macro provides the `set_*_local` functionality and `super.build_phase()` calls the `apply_config_settings()` method under the hood. The `apply_config_settings()` method searches for all appropriate config settings matching this component’s instance and for each match, the appropriate `set_*_local` method is called using the matching `uvm_config_db` setting’s “field_name” and “value”.

The `super.build_phase()` method may be replaced with the `apply_config_settings()` method however it is recommended to use the `super.build_phase()` method.

```
15 class agent extends uvm_agent;
16   int i4;
17   `uvm_component_utils_begin(agent)
18     `uvm_field_int (i4, UVM_ALL_ON)
19   `uvm_component_utils_end
20   virtual function void build_phase(uvm_phase phase);
21     apply_config_settings(1); //No super.build_phase(phase)
22     ...
23   endfunction
24   ...
25 endclass
```

The implicit `get()` method call will not work in the following instances:

- ▶ Missing `uvm_field_*` macro
- ▶ FLAG is set to `UVM_READONLY`
- ▶ Missing `super.build_phase()` or `apply_config_settings()` in `build_phase()`

Below are log messages generated during the simulation phase because of an explicit `apply_config_settings()` function call:

```
UVM_INFO @ 0: env.name_agent_1 [CFGAPL] applying configuration settings
UVM_INFO @ 0: env.name_agent_1 [CFGAPL] applying configuration to field i4
```

To set the value for “i4” of the above agent, env would have the `build_phase()` below:

```
3 function void build_phase (uvm_phase phase);
4   agent_1 = agent::type_id::create("name_agent_1", this);
5   uvm_component_db#(int)::set(this, "name_agent_1", "i4", 1111);
6 endfunction
```

During the build phase of the simulation the agent object's "i4" variable would get value 1111. It is important to note that automatic configuration happens only at build phase.

Command Line

Compilation and simulation time are the major contributors to verification overhead. The ability to change the configuration or parameters without being forced to recompile is critical. The UVM class `uvm_cmdline_processor` provides a mechanism to capture the command line argument and pass to verification components the testcase name, verbosity, configuration and other attributes.

Configuration overriding can only be done from the command line for integer and string using the following:

```
+uvm_set_config_int=<comp>,<field>,<value>
+uvm_set_config_string=<comp>,<field>,<value>
```

There is no way to override the object from the command line, because `uvm_object` cannot be passed to the simulation.

When using the command line argument to set the configuration, make sure that the "<type>" used in `uvm_config_db_set()` and `get()` functions is `uvm_bitstream_t` for integer and the "<type>" for `string` is as shown below:

```
2 class env extends uvm_env;
3   int a;
4   string color;
5   ...
6   ...
7   function new(string name, uvm_component parent);
8     super.new(name, parent);
9   endfunction
10
11  virtual function void build_phase(uvm_phase parent);
12    super.build_phase (phase);
13    if(!uvm_config_db #(uvm_bitstream_t)::get(this, "", "a", a))
14      `uvm_fatal("GET_NOTSUCC", "Get is not successful for a ...");
15    if(!uvm_config_db #(string)::get(this, "", "color", color))
16      `uvm_fatal("GET_NOTSUCC", "Get is not successful for color...");
17    `uvm_info("GET_VALUE", $psprintf ("The value of a = %d and color =
18      %s",a,color),UVM_LOW);
19  endfunction
20  ...
21  endclass
22
23  class test extends uvm_test;
24    int a = 2;
25    string color ="blue";
26    env env_i;
27    ...
28    ...
29    virtual function void build_phase(uvm_phase phase);
30      super.build_phase (phase);
31      env_i = env::type_id::create("env_i", this);
32      uvm_config_db #(uvm_bitstream_t)::set(this, "env_i", "a", a);
33      uvm_config_db #(string)::set(this, "env_i", "color", color);
34    endfunction
35    ...
36  endclass
```

The command line argument for the example above is:

```
<simulation command> +UVM_TESTNAME=test +uvm_set_config_int=uvm_test_top.env_i, a, 6 +uvm_set_config_string=uvm_test_top.env_i, color, red
```

The log message generated during simulation is:

```
UVM_INFO @ 0: reporter [UVM_CMDLINE_PROC] Applying config setting from the
command line: +uvm_set_config_int=uvm_test_top.env_i, a, 6
```

```
UVM_INFO @ 0: reporter [UVM_CMDLINE_PROC] Applying config setting from the
command line: +uvm_set_config_string=uvm_test_top.env_i, color, red
```

Cross-Hierarchical Access

The **set()** and **get()** parameters “cntxt”, “inst_name” and “field_name” make it possible to use a number of different paths to the same object. “cntxt” uses actual object hierarchy whereas “inst_name” and “field_name” uses the hierarchy path with names given to the objects in **create()/new()** method. It is good practice to create the objects with the same name as the object name.

When referencing down in hierarchy, it should be enough to use **this** in “cntxt” and then provide the path and/or names in “inst_name”. “Field_name” should be used just for the name of the object. When referencing upwards in hierarchy, utilize the **uvm_root::get()** function to get access to the hierarchy root, and then reference down from there using “inst_name” parameter.

Figure 5 below clarifies and provides examples how objects can be referenced in **uvm_config_db**.




Figure 5: Options for using “cntxt” and “inst_name” parameters in set() and get()

uvm_config_db does not actually limit how path field name is shared between “cntxt”, “inst_name” and “field_name”. UVM combines all three of these parameters into one “key” that is used to access the database. This feature makes it possible to reference the same object in multiple different ways using the 3 metacharacters *, +, ?. The table below determines the significance of each metacharacter:

Character	Meaning
*	0 or more characters
+	1 or more characters
?	Exactly 1 character

The illustration below shows using these metacharacters for the same object in `uvm_config_db`.




Figure 6: Different path notations to the one and same object

Where To Use – Usage and Its Benefits

Passing Configuration

`uvm_config_db` is used often to configure agents of the testbench and to pass access to signal interfaces. Agent is a class encapsulating sequencer, driver and monitor. Agent usually takes care of generating and receiving data for an interface. The configuration variables or virtual interface are set at agent from top-level and later the agent is responsible for passing the virtual interface or configuration to other sub-components rather than passing it from top-level as shown in the figure below. Agents are often reused either as VIP blocks or across projects. This means that the receiver (`get`) of the information dictates “`type`” and “`field_name`”, and source of the information (`set`) must use proper parameters when setting data into `uvm_config_db`. This is also part of the beauty of `uvm_config_db`: agents can be created without knowing where the parameters or signal interfaces are coming from, from where in the testbench hierarchy the agent object exists, what name it has, or how many instances there are in parallel.




Figure 7: Passing configuration to agent and sub-components

Passing Virtual Interface

Passing the virtual interface across the verification component is the most common requirement when creating the reusable verification environment. The preferred approach of doing this in UVM is to push the virtual interface into the configuration database from top-level. This is because top-level module is not `uvm_component` hence the context is “null” and the instance is the absolute hierarchal path of the component where the virtual interface is assigned.

The absolute hierarchal instance of the component starts with “`uvm_test_top`.” Because the environment is usually instantiated by test and agent, it can extract the virtual interface from the configuration database as shown in the example on the next page

```

2 module tb_top();
3 svt_axi_if vif();
4 ...
5 ...
6 initial begin
7   uvm_config_db #(virtual svt_axi_if)::set(null, "uvm_test_top.env.m_
agent_0", "vif", vif);
8 end
9 endmodule
10
11 class axi_agent extends uvm_agent;
12   virtual svt_axi_if vif();
13   ...
14
15   virtual function _void build_phase(uvm_phase phase);
16     super.build_phase(phase);
17     ...
18     if(!uvm_config_db#(virtual svt_axi_if)::get(this,"","vif", vif))
19       `uvm_fatal("AXI_AGENT:NOVIF", "The virtual interface get is not successful");
20     uvm_config_db#(virtual svt_axi_if)::set(this, "driver","vif",vif);
21     uvm_config_db#(virtual svt_axi_if)::set(this,"monitor","vif",vif);
22   endfunction
23 endclass

```

Event Synchronization

`uvm_config_db` is used to make the object available for others, it does not create new copies of the object. Figure 8 below shows how event-object created by Object A is also made available to Objects X, Y and Z through the `uvm_config_db`. When Object A chooses to use `trigger()` for the event object, others can detect it because they have access to exactly the same object. This demonstrates how the same object "event" is referenced from four different objects with three different instance names.




Figure 8: `uvm_config_db` shares handles to existing object

Some attention needs to be paid that `set()` is called before `get()` for a specific item, otherwise `get()` will fail. Values passed through `uvm_config_db` before `run_phase()` need to take into account that `build_phase()` constructs objects from top to bottom. This is often the desired order, since settings and configurations are usually set from higher levels to lower levels via agents. During the simulation, use of `set()` and `get()` need to be synchronized/timed by the normal testbench operation or by using events to create a synchronization mechanism.

Limitations

`uvm_config_db` can be used anywhere in the hierarchy. The first parameter of `set()` and `get()` functions, "cntxt", however needs to be of type class `uvm_component` (or extended from that). "cntxt" parameter is often given value utilizing class member `this`. So if `set()` or `get()` functions are used outside `uvm_component` extended object, "cntxt" parameter can be given value using `uvm_root::get()`, or just value "null".




Figure 9: set() and get() functions need “cntxt” parameter of type uvm_component or null

One common usage of `uvm_config_db` outside `uvm_components`, is delivering values from `hdl_top` to the testbench, including access to interfaces instantiated on `hdl_top`. Though `hdl_top` is not extended from any UVM class, `uvm_config_db` can still be utilized and communication with the UVM part of the testbench is possible.

If `set()` or `get()` function is used with “cntxt” parameter not pointing to object of `uvm_component` extended classes, there will be a compile error as shown below.

```

25 Error-[SV-IUOT] Illegal us of this
26   test.sv, 9
27   'this' can only be used in a class method.
28
29 Error-[ICTTFC] Incompatible complex type usage
30   test.svh, 281
31   Incompatible complex type usage in task or function call.
32   The following expression is incompatible with the formal parameter of the
33   function. The type of the actual is 'class
34   my_pkg::bus_slave', while the type of the formal is 'class
35   uvm_pkg::uvm_component'. Expression: this
36   Source info: uvm_pkg_uvm_config_db_2116934237::get(this, "\000",
37   this.get_full_name(), my_agent)
  
```

Figure 10: Example error messages when trying to use “this” for non-uvm_component in set/get

Problems, Errors and Debug

Even though operation and use of `set()` and `get()` functions with `uvm_config_db` are logical and quite simple, `uvm_config_db` related debugging is often needed. Some errors may stop the compile or simulation making them easy to find, as opposed to a coding error that simulates without error even though the `get()` function was receiving incorrect objects. Some common types of errors are:

- ▶ Compile time errors
 - Parameter type does not match provided T value
 - Trying to use this-pointer from class not extended from `uvm_component`
- ▶ Simulation time errors
 - `get()` does not find what was set using `set()` due to misspelling of “inst_name” or “field value”
 - `null` object access attributed to `get()` used before `set()`

Synopsys' VCS Discovery Visualization Environment (DVE) has built-in support for UVM debug. Using the GUI, it is possible to get list of "Set calls without Get" and "Get calls without Set". These lists help to find and detect errors in the testbench. Figure 11 below shows the DVE UVM debug dialog window.




Figure 11: Synopsys' VCS DVE UVM debug dialog window

The UVM command line option `+UVM_CONFIG_DB_TRACE` makes all `set()` and `get()` calls visible in the simulation log. However doing this makes the log file too verbose and difficult to interpret. For this reason tracing is typically turned on only when finding a specific `uvm_config_db` problem. Below is an example of log messages printed out when `set()` and `get()` functions are executed.

```
39 UVM_INFO /global/apps4/vcs_2012.09-3/etc/uvm-1.1/base/uvm_resource_db.svh(129) @ 0:
40 reporter [CFGDB/SET]
41 Configuration 'uvm_test_top.env.name_agent_1.i_of_env' (type int)
42 set by uvm_test_top.env.name_agent_1 = (int) 1
43
44 UVM_INFO /global/apps4/vcs_2012.09-3/etc/uvm-1.1/base/uvm_resource_db.svh(129) @ 0;
45 reporter [CFGDB/GET]
46 Configuration 'uvm_test_top.env.name_agent_1.i_of_env' (type int)
47 read by uvm_test_top.env.name_agent_1 = (int) 1
```

Sometimes it may help just to print out the name of the object. UVM object has functions `get_name()` and `get_full_name()`. By using these, it can be verified manually that names used in the source code and named objects at runtime match. Below is an example of how to print the object's name.

```
49 $display("this.get_name=%0s, this.get_full_name= %0s", this.get_name(),
this.get_full_name());
```

Conclusion

When using UVM you can't avoid `uvm_config_db`. So it's better to get a solid understanding about what the `set()` and `get()` functions of the `uvm_config_db` do and how you can use them more efficiently in building your testbench. Below are some do's and don'ts found to be useful when using UVM.

Do's:

- ▶ For simplicity and to avoid confusion, use the "field name" as the variable name
- ▶ Investigate an upshot warning/error on an unsuccessful `get()` method call
- ▶ Set the configuration variable needed across the verification component in the agent's test environment to enable the agent to later set its sub-components

Don'ts:

- ▶ Avoid using the `uvm_config_db` mechanism excessively as it may cause performance issues
- ▶ Avoid using the automatic configuration or implicit `get()` method call

Apart from the above recommendations, it is recommended to use a UVM-aware GUI-based debugging tool such as Synopsys' VCS Discovery Visualization Environment (DVE). As part of Synopsys Professional Services we have used these concepts across multiple customer engagements to successfully deploy UVM. For more information on these services, see www.synopsys.com/services.

Appendix

Below is a sample UVM environment showcasing the examples presented earlier. `set()` and `get()` functions utilize only integers (`int`) though classes and interfaces that would normally be used.

```
1 // Usage
2 // vcs -R -sverilog -ntb_opts uvm-1.1 -debug_all +vcs+vcdpluson
-1 sim.log
3 // -q example.sv +UVM_TESTNAME=test_a
4 import uvm_pkg::*;
5
6 module dut;
7     int dut_int = 10;
8     initial uvm_config_db#(int)::set(uvm_root::get(),
9                                     "",
10                                    "from_dut",
11                                    dut_int
12                                    );
13 endmodule
14
15 module top;
16     initial run_test();
17     dut i_dut();
18 endmodule
19
20 class agent extends uvm_component;
21     int i1_agent, i2_agent, i3_agent; // to receive values from uvm_config_db
22     int i4; // to use automatic configuration
23     uvm_event new_values; // to signal new step in test (env->agent)
24     `uvm_component_utils_begin (agent)
25     `uvm_field_int(i4, UVM_ALL_ON)
26     `uvm_component_utils_end
27     function new (string name, uvm_component parent);
28         super.new(name, parent);
29     endfunction
30     function void build_phase (uvm_phase phase);
31         super.build_phase(phase);
32         uvm_config_db#(uvm_event)::get(this, "", "new_values", new_values);
33         if (new_values == null)
34             `uvm_fatal(get_name(),
35                        "new_values must be set in uvm_config_db"
36                        );
37         $display(" Agent \"%0s\" got Automatic configuration: i4=%0d",
38                get_name(),
39                i4
40                );
41     endfunction
42     task run_phase (uvm_phase phase);
43         while (1)
44             begin
45                 uvm_config_db#(int)::get(this,
46                                           "",
47                                           "i_of_env",
48                                           i1_agent
49                                           );
```

```

50         uvm_config_db#(int)::get(uvm_root::get(),
51             "uvm_test_top.env.name_agent_1",
52             "i_of_env",
53             i2_agent
54         );
55         $display(" Agent \"%0s\" got: %0d (and stole %0d from agent1)",
56             get_name(),
57             i1_agent,
58             i2_agent
59         );
60         uvm_config_db#(int)::get(this, "", "from_dut", i3_agent);
61         $display(" Agent \"%0s\" got %0d from DUT",
62             get_name(),
63             i3_agent
64         );
65         new_values.wait_trigger();
66
67         end
68     endtask
69 endclass
70
71 class env extends uvm_env;
72     `uvm_component_utils(env)
73     agent agent_1, agent_2, agent_3;
74     int i1_env=1, i2_env=2, i3_env=3,
75         i4_env=4, i5_env=5, i6_env=6,
76         i7_env=7, i8_env=8;
77     uvm_event new_values;
78     function new(string name, uvm_component parent);
79         super.new(name, parent);
80     endfunction
81     function void build_phase (uvm_phase phase);
82         agent_1 = agent::type_id::create("name_agent_1, this);
83         agent_2 = agent::type_id::create("name_agent_2, this);
84         agent_3 = agent::type_id::create("name_agent_3, this);
85         set_config_int("name_agent_1", "i4", 1111);
86         set_config_int("name_agent_2", "i4", 2222);
87         set_config_int("name_agent_3", "i4", 3333);
88         new_values = new("new_values");
89         //Share event through uvm_config_db with agents
90         uvm_config_db#(uvm_event)::set(this,
91             "name_agent_*",
92             "new_values",
93             new_values
94         );
95     endfunction
96     task run_phase (uvm_phase phase);
97         phase.raise_objections(this);
98         //uvm_config: share data with one specific object
99         $display(" --- 1, 2, 3 to every agent separately --- ");
100        uvm_config_db#(int)::set(agent_1,
101            "",
102            "i_of_env",
103            i1_env
104        );
105        uvm_config_db#(int)::set(this,
106            "name_agent_2",
107            "i_of_env",
108            i2_env
109        );
110        uvm_config_db#(int)::set(uvm_root::get(),
111            "uvm_test_top.env.name_agent_3",
112            "i_of_env",
113            i3_env
114        );
115        // uvm_config_db: share data with multiple objects using regexp

```

```

116     new_values.trigger(); #1;
117     $display(" --- 4 to every agent, regexp name _agent_? --- ");
118     uvm_config_db#(int)::set(this,
119         "name_agent_?",
120         "i_of_env",
121         i4_env
122     );
123     new_values.trigger(); #1;
124     $display(" --- 5 to every agent, regexp name _agent* --- ");
125     uvm_config_db#(int)::set(this,
126         "name_agent_*",
127         "i_of_env",
128         i5_env
129     );
130     new_values.trigger(); #1;
131     $display(" --- 6 to every agent, regexp *agent* --- ");
132     uvm_config_db#(int)::set(uvm_root::get(),
133         "*agent*",
134         "i_of_env",
135         i6_env
136     );
137     // uvm_config_db: share data with everyone
138     new_values.trigger(); #1;
139     $display(" --- 7 to everyone, regexp *--- ");
140     uvm_config_db#(int)::set(uvm_root::get(),
141         "*",
142         "i_of_env",
143         i7_env
144     );
145     new_values.trigger(); #1;
146     $display(" --- 8 to everyone, regexp *--- ");
147     uvm_config_db#(int)::set(null,
148         "*",
149         "i_of_env",
150         i8_env
151     );
152     new_values.trigger();
153     phase.drop_objection(this);
154 endtask
155 endclass
156
157 class test_a extends uvm_test;
158     `uvm_component_utils (test_a)
159     env env;
160     function new (string name="test_a", uvm_component parent=null);
161         super.new (name, parent);
162         env = new("env", this);
163     endfunction
164     function void end_of_elaboration();
165         print();
166     endfunction
167     task run_phase(uvm_phase phase);
168         #1000; global_stop_request();
169     endtask
170 endclass

```

References

¹ Accellera Systems Initiative Universal Verification Methodology (UVM) 1.1 User's Guide, May 18, 2011

² Accellera Systems Initiative Universal Verification Methodology (UVM) 1.1 Class Reference Manual, June 2011